Yaounde, March 23, 2005

Wolfram Koepf: Computer Algebra Algorithms for Orthogonal Polynomials and Special Functions

```
> restart;
```

Computation of Power Series

Maple supports truncated power series

```
> series(exp(x),x);
```

$$1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+O(x^6)$$

The following algorithm for the computation of Formal Power Series is from
Koepf, Wolfram: Power Series in Computer Algebra, Journal of Symbolic Computation 13, 1992, 581-603

```
> read "FPS.mpl";
```

Package Formal Power Series, Maple V - Maple 8

Copyright 1995, Dominik Gruntz, University of Basel

Copyright 2002, Detlef Müller & Wolfram Koepf, University of Kassel

```
> FPS(exp(x),x);
```

$$\sum_{k=0}^{\infty}\frac{x^k}{k!}$$

```
> infolevel[FPS]:=5:
> FPS(exp(x),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   DE of degree   1   found.
FPS/FPS:   DE =
```

$$F'(x)-F(x)=0$$

```
FPS/FPS:   RE =
```

$$a(k+1)=\frac{a(k)}{k+1}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    1
FPS/hypergeomRE:   RE:
```

$$(k+1)\,a(k+1)=a(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    0
FPS/hypergeomRE:   a(0) =   1
```

$$\sum_{k=0}^{\infty}\frac{x^k}{k!}$$

```
> FPS(exp(x^2),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   DE of degree   1   found.
FPS/FPS:   DE =
```

$$F'(x)-2\,x\,F(x)=0$$

```
FPS/FPS:   RE =
```

$$a(k+1) = \frac{2\, a(k-1)}{k+1}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    2
FPS/hypergeomRE:   RE:
```

$$(k+2)\, a(k+2) = 2\, a(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    -1
FPS/hypergeomRE:   a(0) =   1
```

$$\sum_{k=0}^{\infty} \frac{x^{(2k)}}{k!}$$

a Puiseux series

```
> FPS(exp(sqrt(x)),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   looking for DE of degree   2
FPS/FPS:   DE of degree   2   found.
FPS/FPS:   DE =
```

$$4\, x\, F''(x) + 2\, F'(x) - F(x) = 0$$

```
FPS/FPS:   RE =
```

$$a(k+1) = \frac{1}{2} \frac{a(k)}{(k+1)\,(2\,k+1)}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    1
FPS/hypergeomRE:   RE:
```

$$2\,(k+1)\,(2\,k+1)\, a(k+1) = a(k)$$

```
FPS/hypergeomRE:   RE modified to    k = 1/2*k
FPS/hypergeomRE:   => f :=    exp(x)
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    2
FPS/hypergeomRE:   RE:
```

$$(k+2)\,(k+1)\, a(k+2) = a(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    0
FPS/hypergeomRE:   a(0) =   1
FPS/hypergeomRE:   a(1) =   1
```

$$\left(\sum_{k=0}^{\infty} \frac{x^k}{(2\,k)!}\right) + \left(\sum_{k=0}^{\infty} \frac{x^{(k+1/2)}}{(2\,k+1)!}\right)$$

```
> FPS(arcsin(x),x);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   looking for DE of degree   2
FPS/FPS:   DE of degree   2   found.
FPS/FPS:   DE =
```

$$(-1 + x^2)\, F''(x) + x\, F'(x) = 0$$

```
FPS/FPS:   RE =
```

$$a(k+2) = \frac{k^2\, a(k)}{(k+1)\,(k+2)}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    2
FPS/hypergeomRE:   RE:
```

$$-(k+1)\,(k+2)\, a(k+2) = -k^2\, a(k)$$

```
FPS/hypergeomRE:   RE valid for all k >=    0
FPS/hypergeomRE:   a(0) =   0
```

```
FPS/hypergeomRE:   a(2*j) = 0   for all j>0.
FPS/hypergeomRE:   a(1) =   1
```

$$\sum_{k=0}^{\infty} \frac{(2\,k)!\,4^{(-k)}\,x^{(2\,k+1)}}{(k!)^2\,(2\,k+1)}$$

```
> infolevel[FPS]:=0:
```

procedure to compute a holonomic differential equation

```
> DE:=HolonomicDE(arcsin(x),F(x));
```

$$DE := (x-1)\,(x+1)\left(\frac{d^2}{dx^2}\,\mathrm{F}(x)\right)+x\left(\frac{d}{dx}\,\mathrm{F}(x)\right)=0$$

```
> dsolve(DE,F(x));
```

$$\mathrm{F}(x)=\_C1+\ln(x+\sqrt{-1+x^2}\,)\,\_C2$$

some final examples: a Laurent series

```
> FPS(arcsin(x)^2/x^5,x);
```

$$\sum_{k=0}^{\infty} \frac{(k!)^2\,4^k\,x^{(2\,k-3)}}{(1+2\,k)!\,(k+1)}$$

a complicated example that cannot be found in Gradshteyn/Ryshik

```
> FPS(exp(arcsin(x)),x);
```

$$\left(\sum_{k=0}^{\infty} \frac{\left(\prod_{j=0}^{k}(4\,j^2+1)\right)x^{(2\,k)}}{(4\,k^2+1)\,(2\,k)!}\right)+\left(\sum_{k=0}^{\infty} \frac{\left(\prod_{j=0}^{k}(1+2\,j+2\,j^2)\right)2^k\,x^{(2\,k+1)}}{(2\,k+1)!\,(2\,k^2+2\,k+1)}\right)$$

and an asymptotic series

```
> FPS((erf(x)-1)*exp(x^2),x=infinity);
```

$$-\frac{\displaystyle\sum_{k=0}^{\infty} \frac{(\text{-}1)^k\,(2\,k)!\,4^{(-k)}\left(\frac{1}{x}\right)^{(2\,k+1)}}{k!}}{\sqrt{\pi}}$$

Also covered are holonomic special functions

```
> FPS(LegendreP(n,x),x);
```

$$\frac{2\sqrt{\pi}\left(\displaystyle\sum_{k=0}^{\infty} \frac{\mathrm{pochhammer}\left(-\frac{n}{2},k\right)\mathrm{pochhammer}\left(\frac{n}{2}+\frac{1}{2},k\right)4^k\,x^{(2\,k)}}{(2\,k)!}\right)}{\Gamma\left(\frac{1}{2}-\frac{n}{2}\right)\Gamma\left(\frac{n}{2}\right)n}-\frac{2\sqrt{\pi}\left(\displaystyle\sum_{k=0}^{\infty} \frac{\mathrm{pochhammer}\left(\frac{1}{2}-\frac{n}{2},k\right)\mathrm{pochhammer}\left(1+\frac{n}{2},k\right)4^k\,x^{(1+2\,k)}}{(1+2\,k)!}\right)}{\Gamma\left(\frac{n}{2}+\frac{1}{2}\right)\Gamma\left(-\frac{n}{2}\right)}$$

```
> FPS(LegendreP(n,x),x=1);
```

$$\sum_{k=0}^{\infty} \frac{(-1)^k \, 2^{(-k)} \, \text{pochhammer}(n+1, k) \, \text{pochhammer}(-n, k) \, (x-1)^k}{(k!)^2}$$

```
> HolonomicDE(LegendreP(n,x),F(x));
```

$$(-1+x)(x+1)\left(\frac{d^2}{dx^2} F(x)\right) - (n+1)\, n\, F(x) + 2\, x \left(\frac{d}{dx} F(x)\right) = 0$$

```
>
```

Algebra of Holonomic Functions

```
> with(gfun);
```

[*Laplace, algebraicsubs, algeqtodiffeq, algeqtoseries, algfuntoalgeq, borel, cauchyproduct, diffeq*diffeq, diffeq+diffeq, diffeqtable, diffeqtohomdiffeq, diffeqtorec, guesseqn, guessgf, hadamardproduct, holexprtodiffeq, invborel, listtoalgeq, listtodiffeq, listtohypergeom, listtolist, listtoratpoly, listtorec, listtoseries, maxdegcoeff, maxdegeqn, maxordereqn, mindegcoeff, mindegeqn, minordereqn, optionsgf, poltodiffeq, poltorec, ratpolytocoeff, rec*rec, rec+rec, rectodiffeq, rectohomrec, rectoproc, seriestoalgeq, seriestodiffeq, seriestohypergeom, seriestolist, seriestoratpoly, seriestorec, seriestoseries*]

We consider the function sin(x)*exp(x):

The differential equation of sin(x):

```
> DE1:=diff(F(x),x$2)+F(x)=0;
```

$$DE1 := \left(\frac{d^2}{dx^2} F(x)\right) + F(x) = 0$$

The differential equation of exp(x):

```
> DE2:=diff(F(x),x)-F(x)=0;
```

$$DE2 := \left(\frac{d}{dx} F(x)\right) - F(x) = 0$$

```
> `diffeq*diffeq`(DE1,DE2,F(x));
```

$$\left(\frac{d^2}{dx^2} F(x)\right) - 2\left(\frac{d}{dx} F(x)\right) + 2\, F(x)$$

and the sum sin(x)+exp(x) satisfies

```
> `diffeq+diffeq`(DE1,DE2,F(x));
```

$$\left(\frac{d^3}{dx^3} F(x)\right) - \left(\frac{d^2}{dx^2} F(x)\right) + \left(\frac{d}{dx} F(x)\right) - F(x)$$

Now a more complicated example: exp(x)*Ai(x)

```
> DE1:=diff(F(x),x)-F(x)=0;
```

$$DE1 := \left(\frac{d}{dx} F(x)\right) - F(x) = 0$$

```
> DE2:=HolonomicDE(AiryAi(x),F(x));
```

$$DE2 := \left(\frac{d^2}{dx^2} F(x)\right) - x\, F(x) = 0$$

```
> `diffeq*diffeq`(DE1,DE2,F(x));
```

$$(-x+1)\,F(x)+\left(\frac{d^2}{dx^2}F(x)\right)-2\left(\frac{d}{dx}F(x)\right)$$

and the sum exp(x)+Ai(x) satisfies

```
> `diffeq+diffeq`(DE1,DE2,F(x));
```

$$\{(1-x+x^2)\,F(x)+(-x^2+x)\left(\frac{d}{dx}F(x)\right)-x\left(\frac{d^2}{dx^2}F(x)\right)+(x-1)\left(\frac{d^3}{dx^3}F(x)\right),$$
$$(D^{(2)})(F)(0)=\_C_0\}$$

Similarly, HolonomicDE yields

```
> HolonomicDE(exp(x)+AiryAi(x),F(x));
```

$$(x-1)\left(\frac{d^3}{dx^3}F(x)\right)+(1-x+x^2)\,F(x)-x\left(\frac{d^2}{dx^2}F(x)\right)-x\,(x-1)\left(\frac{d}{dx}F(x)\right)=0$$

```
>
```

Hypergeometric Functions

```
> simplify(x*hypergeom([],[3/2],-x^2/4));
```

$$\sin(x)$$

```
> hypergeom([a,b],[c],x);
```

$$\text{hypergeom}([a,b],[c],x)$$

```
> sumtools[hyperterm]([a,b],[c],x,k);
```

$$\frac{\text{pochhammer}(a,k)\,\text{pochhammer}(b,k)\,x^k}{\text{pochhammer}(c,k)\,k!}$$

```
> sum(sumtools[hyperterm]([a,b],[c],x,k),k=0..infinity);
```

$$\text{hypergeom}([a,b],[c],x)$$

```
> hypergeom([a,b],[c],1);
```

$$\text{hypergeom}([a,b],[c],1)$$

```
> simplify(hypergeom([a,b],[c],1));
```

$$\frac{\Gamma(c)\,\Gamma(c-a-b)}{\Gamma(c-a)\,\Gamma(c-b)}$$

```
>
```

Identification of Hypergeometric Functions

We are interested in

```
> s:=Sum((-1)^k/(2*k+1)!*x^(2*k+1),k=0..infinity);
```

$$s:=\sum_{k=0}^{\infty}\frac{(-1)^k\,x^{(2k+1)}}{(2k+1)!}$$

```
> F:=k->(-1)^k/(2*k+1)!*x^(2*k+1);
```

$$F:=k\to\frac{(-1)^k\,x^{(2k+1)}}{(2k+1)!}$$

```
> r:=F(k+1)/F(k);
```

$$r := \frac{(-1)^{(k+1)} x^{(2k+3)} (2k+1)!}{(2k+3)! (-1)^k x^{(2k+1)}}$$

```
> expand(r);
```

$$-\frac{x^2}{(2k+2)(2k+3)}$$

Hence

```
> s=F(0)*hypergeom([],[3/2],-x^2/4);
```

$$\sum_{k=0}^{\infty} \frac{(-1)^k x^{(2k+1)}}{(2k+1)!} = x \operatorname{hypergeom}\left([\ ], \left[\frac{3}{2}\right], -\frac{x^2}{4}\right)$$

The following procedure uses the given algorithm and gives therefore the hypergeometric form:

```
> sumtools[Sumtohyper](F(k),k);
```

$$x \operatorname{Hypergeom}\left([\ ], \left[\frac{3}{2}\right], -\frac{x^2}{4}\right)$$

Another example

```
> F:=binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k;
```

$$F := \operatorname{binomial}(n, k) \operatorname{binomial}(-n-1, k) \left(-\frac{x}{2} + \frac{1}{2}\right)^k$$

```
> Sum(F,k=0..n)=sumtools[Sumtohyper](F,k);
```

$$\sum_{k=0}^{n} \operatorname{binomial}(n, k) \operatorname{binomial}(-n-1, k) \left(-\frac{x}{2} + \frac{1}{2}\right)^k =$$

$$\operatorname{Hypergeom}\left([-n, n+1], [1], -\frac{x}{2} + \frac{1}{2}\right)$$

Details of this algorithm and an implementation can be found in the book
Wolfram Koepf: *Hypergeometric Summation*, Vieweg, Braunschweig/Wiesbaden, 1998

```
>
```

Computation of Recurrence Equations for Hypergeometric Functions: Faasenmyer's Algorithm

How does one generate the result

```
> Sum(binomial(n,k),k=0..n)=
  sum(binomial(n,k),k=0..n);
```

$$\sum_{k=0}^{n} \operatorname{binomial}(n, k) = 2^n$$

We do the following more complicated example with Maple:

```
> Sum(k*binomial(n,k),k=0..n)=
  sum(k*binomial(n,k),k=0..n);
```

$$\sum_{k=0}^{n} k \operatorname{binomial}(n, k) = \frac{2^n n}{2}$$

```
> F:=(n,k)->k*binomial(n,k);
```

$$F := (n, k) \to k\,\mathrm{binomial}(n, k)$$

```
> ansatz:=sum(sum(a(j,i)*F(n+j,k+i),i=0..1),j=0..1);
```

$$ansatz := \mathrm{a}(0,0)\,k\,\mathrm{binomial}(n,k) + \mathrm{a}(0,1)\,(k+1)\,\mathrm{binomial}(n,k+1) + \mathrm{a}(1,0)\,k\,\mathrm{binomial}(n+1,k) + \mathrm{a}(1,1)\,(k+1)\,\mathrm{binomial}(n+1,k+1)$$

```
> ansatz:=ansatz/F(n,k);
```

$$ansatz := (\mathrm{a}(0,0)\,k\,\mathrm{binomial}(n,k) + \mathrm{a}(0,1)\,(k+1)\,\mathrm{binomial}(n,k+1) + \mathrm{a}(1,0)\,k\,\mathrm{binomial}(n+1,k) + \mathrm{a}(1,1)\,(k+1)\,\mathrm{binomial}(n+1,k+1)) / (k\,\mathrm{binomial}(n,k))$$

```
> ansatz:=expand(ansatz);
```

$$ansatz := \mathrm{a}(0,0) + \frac{\mathrm{a}(0,1)\,n}{k+1} - \frac{k\,\mathrm{a}(0,1)}{k+1} + \frac{\mathrm{a}(0,1)\,n}{k\,(k+1)} - \frac{\mathrm{a}(0,1)}{k+1} + \frac{\mathrm{a}(1,0)\,n}{n-k+1} + \frac{\mathrm{a}(1,0)}{n-k+1} + \frac{\mathrm{a}(1,1)\,n}{k+1} + \frac{\mathrm{a}(1,1)}{k+1} + \frac{\mathrm{a}(1,1)\,n}{k\,(k+1)} + \frac{\mathrm{a}(1,1)}{k\,(k+1)}$$

```
> ansatz:=normal(ansatz);
```

$$ansatz := (-k^2\,\mathrm{a}(0,0) + k^2\,\mathrm{a}(0,1) + \mathrm{a}(0,0)\,k\,n - 2\,\mathrm{a}(0,1)\,n\,k - \mathrm{a}(1,1)\,k - \mathrm{a}(1,1)\,n\,k + \mathrm{a}(1,0)\,n\,k + \mathrm{a}(1,0)\,k + \mathrm{a}(0,0)\,k - k\,\mathrm{a}(0,1) + \mathrm{a}(1,1)\,n^2 + \mathrm{a}(0,1)\,n^2 + \mathrm{a}(0,1)\,n + 2\,\mathrm{a}(1,1)\,n + \mathrm{a}(1,1)) / ((n-k+1)\,k)$$

```
> ansatz:=numer(ansatz);
```

$$ansatz := -k^2\,\mathrm{a}(0,0) + k^2\,\mathrm{a}(0,1) + \mathrm{a}(0,0)\,k\,n - 2\,\mathrm{a}(0,1)\,n\,k - \mathrm{a}(1,1)\,k - \mathrm{a}(1,1)\,n\,k + \mathrm{a}(1,0)\,n\,k + \mathrm{a}(1,0)\,k + \mathrm{a}(0,0)\,k - k\,\mathrm{a}(0,1) + \mathrm{a}(1,1)\,n^2 + \mathrm{a}(0,1)\,n^2 + \mathrm{a}(0,1)\,n + 2\,\mathrm{a}(1,1)\,n + \mathrm{a}(1,1)$$

```
> eqs:={coeffs(ansatz,k)};
```

$$eqs := \{2\,\mathrm{a}(1,1)\,n + \mathrm{a}(1,1) + \mathrm{a}(1,1)\,n^2 + \mathrm{a}(0,1)\,n^2 + \mathrm{a}(0,1)\,n,\ -\mathrm{a}(1,1)\,n + \mathrm{a}(0,0)\,n - 2\,\mathrm{a}(0,1)\,n - \mathrm{a}(1,1) - \mathrm{a}(0,1) + \mathrm{a}(1,0)\,n + \mathrm{a}(1,0) + \mathrm{a}(0,0),\ -\mathrm{a}(0,0) + \mathrm{a}(0,1)\}$$

```
> sol:=solve(eqs,{seq(seq(a(j,i),j=0..1),i=0..1)});
```

$$sol := \{\mathrm{a}(1,0) = 0,\ \mathrm{a}(0,0) = -\frac{(n+1)\,\mathrm{a}(1,1)}{n},\ \mathrm{a}(0,1) = -\frac{(n+1)\,\mathrm{a}(1,1)}{n},\ \mathrm{a}(1,1) = \mathrm{a}(1,1)\}$$

```
> re:=sum(sum(a(j,i)*f(n+j,k+i),i=0..1),j=0..1);
```

$$re := \mathrm{a}(0,0)\,\mathrm{f}(n,k) + \mathrm{a}(0,1)\,\mathrm{f}(n,k+1) + \mathrm{a}(1,0)\,\mathrm{f}(n+1,k) + \mathrm{a}(1,1)\,\mathrm{f}(n+1,k+1)$$

```
> re:=subs(sol,re);
```

$$re := -\frac{(n+1)\,\mathrm{a}(1,1)\,\mathrm{f}(n,k)}{n} - \frac{(n+1)\,\mathrm{a}(1,1)\,\mathrm{f}(n,k+1)}{n} + \mathrm{a}(1,1)\,\mathrm{f}(n+1,k+1)$$

```
> re:=numer(normal(re/a(1,1)));
```

$$re := -\mathrm{f}(n,k)\,n - \mathrm{f}(n,k) - \mathrm{f}(n,k+1)\,n - \mathrm{f}(n,k+1) + \mathrm{f}(n+1,k+1)\,n$$

```
> s:='s':
> RE:=subs({seq(seq(f(n+j,k+i)=s(n+j),i=0..1),j=0..1)},re);
```

$$RE := -2\,\mathrm{s}(n)\,n - 2\,\mathrm{s}(n) + \mathrm{s}(n+1)\,n$$

```
> RE:=map(factor,collect(RE,s))=0;
```

$$RE := -2\,(n+1)\,s(n) + s(n+1)\,n = 0$$

Now we use the implementation from the book
Wolfram Koepf: *Hypergeometric Summation*, Vieweg, Braunschweig/Wiesbaden, 1998

```
> restart; read "hsum9.mpl";
```

Package "Hypergeometric Summation", Maple V - Maple 9

Copyright 1998-2004, Wolfram Koepf, University of Kassel

```
> fasenmyer(k*binomial(n,k),k,s(n),1,1);
```

$$n\,s(n+1) - 2\,s(n)\,(n+1) = 0$$

```
> fasenmyer(binomial(n,k)^2,k,s(n),1,1);
Error, (in kfreerec) No kfree recurrence equation of order (1,1) exists
```

```
> fasenmyer(binomial(n,k)^2,k,s(n),2,1);
```

$$(n+2)\,s(n+2) - 2\,s(n+1)\,(2\,n+3) = 0$$

```
> fasenmyer(binomial(n-k,k),k,s(n),2,1);
```

$$s(n+2) - s(n) - s(n+1) = 0$$

```
> [seq(sum(binomial(n-k,k),k=0..n),n=0..10)]; n:='n':
```

$$[1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89]$$

```
> fasenmyer((-1)^k*binomial(n,k)^2,k,s(n),2,2);
```

$$(n+2)\,s(n+2) + 4\,s(n)\,(n+1) = 0$$

```
> fasenmyer(binomial(n,k)^3,k,s(n),2,1);
Error, (in kfreerec) No kfree recurrence equation of order (2,2) exists
```

```
> fasenmyer(binomial(n,k)^3,k,s(n),3,1);
```

$$(3\,n+4)\,(n+3)^2\,s(n+3) - 2\,(9\,n^3 + 57\,n^2 + 116\,n + 74)\,s(n+2) - (3\,n+5)\,(15\,n^2 + 55\,n + 48)\,s(n+1) - 8\,(3\,n+7)\,(n+1)^2\,s(n) = 0$$

Zeilberger's Algorithm

```
> sumrecursion(k*binomial(n,k),k,s(n));
```

$$2\,(n+1)\,s(n) - s(n+1)\,n = 0$$

```
> sumrecursion((-1)^k*binomial(n,k)^2,k,s(n));
```

$$(n+2)\,s(n+2) + 4\,(n+1)\,s(n) = 0$$

```
> sumrecursion(binomial(n,k)^3,k,s(n));
```

$$8\,(n+1)^2\,s(n) + (7\,n^2 + 21\,n + 16)\,s(n+1) - (n+2)^2\,s(n+2) = 0$$

With Zeilberger's algorithm, we can do more complicated examples.

The Apéry numbers

```
> Sum(binomial(n,k)^2*binomial(n+k,k)^2,k=0..n);
```

$$\sum_{k=0}^{n} \mathrm{binomial}(n,k)^2\,\mathrm{binomial}(n+k,k)^2$$

satisfy the recurrence equation

```
> sumrecursion(binomial(n,k)^2*binomial(n+k,k)^2,k,A(n));
```

$$(n+1)^3\,A(n) - (2\,n+3)\,(17\,n^2 + 51\,n + 39)\,A(n+1) + (n+2)^3\,A(n+2) = 0$$

The power sums of the binomial coefficients were worth a paper in the 1980s:

```
> sumrecursion(binomial(n,k)^4,k,s(n));
```

$$4\,(4\,n+5)\,(4\,n+3)\,(n+1)\,\mathrm{s}(n)+2\,(2\,n+3)\,(3\,n^2+9\,n+7)\,\mathrm{s}(n+1)$$
$$-\,(n+2)^3\,\mathrm{s}(n+2)=0$$

```
> sumrecursion(binomial(n,k)^5,k,s(n));
```

$$32\,(55\,n^2+253\,n+292)\,(n+1)^4\,\mathrm{s}(n)-$$
$$(514048+2682770\,n^2+2082073\,n^3+900543\,n^4+205799\,n^5+19415\,n^6+1827064\,n)$$
$$\mathrm{s}(n+1)-$$
$$(310827\,n^2+205949\,n^3+75498\,n^4+79320+245586\,n+14553\,n^5+1155\,n^6)\,\mathrm{s}(n+2)$$
$$+\,(55\,n^2+143\,n+94)\,(n+3)^4\,\mathrm{s}(n+3)=0$$

```
> sumrecursion(binomial(n,k)^6,k,s(n));
```

$$24\,(6\,n+5)\,(2\,n+3)\,(6\,n+7)\,(91\,n^3+637\,n^2+1491\,n+1167)\,(n+1)^3\,\mathrm{s}(n)-($$
$$22934340+280311768\,n^2+378741807\,n^3+327503034\,n^4+187916733\,n^5+153881\,n^9$$
$$+\,71536002\,n^6+17419983\,n^7+2462096\,n^8+120507876\,n)\,\mathrm{s}(n+1)-(n+2)\,(3458\,n^8$$
$$+\,57057\,n^7+408555\,n^6+1656761\,n^5+4158211\,n^4+6610054\,n^3+6496560\,n^2$$
$$+\,3609252\,n+868140)\,\mathrm{s}(n+2)$$
$$+\,(n+2)\,(91\,n^3+364\,n^2+490\,n+222)\,(n+3)^5\,\mathrm{s}(n+3)=0$$

```
> sumrecursion(binomial(n,k)^7,k,s(n));
```

$$128\,(427721\,n^8+9776480\,n^7+97373115\,n^6+551893883\,n^5+1946706314\,n^4$$
$$+\,4375566933\,n^3+6119692458\,n^2+4869142152\,n+1687389120)\,(n+2)^2\,(n+1)^6$$
$$\mathrm{s}(n)-(2193807069981696+3244263785\,n^{16}+198784165636833\,n^{12}$$
$$+\,1132823172700850\,n^{11}+126062821360\,n^{15}+2283968506414\,n^{14}$$
$$+\,17662464938922851\,2\,n^5+54690808998655008\,n^2+11479132240163246 4\,n^3$$
$$+\,166377205614902736\,n^4+142107402452328480\,n^6+88420368230599884\,n^7$$
$$+\,16415798739266369\,n^9+4900968186516568\,n^{10}+43010799826545440\,n^8$$
$$+\,25606027648545\,n^{13}+16071328274727552\,n)\,\mathrm{s}(n+2)-(112552666603632\,n^2$$
$$+\,198216442561728\,n^3+236869167238448\,n^4+30368191\,n^{14}+203258395972016\,n^5$$
$$+\,6117625957887\,n^9+1239681510073\,n^{10}+129212210111012\,n^6+61845130443640\,n^7$$
$$+\,22406262825083\,n^8+6127621340928+180879396742\,n^{11}+17971912105\,n^{12}$$
$$+\,1088916563\,n^{13}+38805627231072\,n)\,(n+3)^2\,\mathrm{s}(n+3)+(45209280+209877096\,n$$
$$+\,421557546\,n^2+478442631\,n^3+335597294\,n^4+149008897\,n^5+40913943\,n^6$$
$$+\,6354712\,n^7+427721\,n^8)\,(n+3)^2\,(n+4)^6\,\mathrm{s}(n+4)+(671258737065984$$
$$+\,1496821367706988 8\,n^2+29207641278240480\,n^3+38801484010527532\,n^4$$
$$+\,15821827511\,n^{14}+37123771902845896\,n^5+1764446202422005\,n^9$$
$$+\,402186441422282\,n^{10}+26380423880989287\,n^6+14144725417173505\,n^7$$
$$+\,5750836202090468\,n^8+66049812430419\,n^{11}+7388757320392\,n^{12}$$
$$+\,504038219279\,n^{13}+4674653721868800\,n)\,(n+2)^2\,\mathrm{s}(n+1)=0$$

```
> sumrecursion(binomial(n,k)^8,k,s(n));
```

$$16\,(8\,n+13)\,(8\,n+7)\,(8\,n+9)\,(8\,n+11)\,(n+2)\,(102375360\,n^{11}+3186433080\,n^{10}$$

$$+ 44960611518\, n^9 + 379608257007\, n^8 + 2130886001250\, n^7 + 8350001129322\, n^6$$
$$+ 23306855546382\, n^5 + 46339428278457\, n^4 + 64315605847158\, n^3$$
$$+ 59346884858090\, n^2 + 32767840545852\, n + 8201727801720)\, (n + 1)^5\, \mathrm{s}(n) - 12\, ($$
$$113072227158706427536 8\, n^3 + 53113860263695806\, n^{17}$$
$$+ 11045645792318410061 48\, n^{10} + 35917391085875025960 80\, n^5$$
$$+ 13131083335252556274\, n^{14} + 2325420945730194232698\, n^4 + 858467294792387280 0$$
$$+ 531359186174014492 89\, n^{13} + 3251191961324982788923\, n^8 + 334201973882040\, n^{19}$$
$$+ 2084807859419201931337\, n^9 + 839255104964831077 44\, n + 280508486400\, n^{21}$$
$$+ 14060487880800\, n^{20} + 4161525693270481443599\, n^7 + 4325426980028204773202\, n^6$$
$$+ 3882809750612836159 68\, n^2 + 26371809868650853 74\, n^{15} + 4235816801138059 17\, n^{16}$$
$$+ 1765911402240850944 02\, n^{12} + 4850739466331070894 11\, n^{11} + 5009309431465140\, n^{18}$$
$$)\, \mathrm{s}(n + 2) - 2\, (828656447429098560\, n^2 + 2022592897697984984\, n^3$$
$$+ 3438871758751753182\, n^4 + 770715264100878\, n^{14} + 4325192582660019738\, n^5$$
$$+ 9119302147464052 78\, n^9 + 352423811632001922\, n^{10} + 310658029920\, n^{17}$$
$$+ 6961524480\, n^{18} + 4170731594507388838\, n^6 + 3153073563533903151\, n^7$$
$$+ 1894748039012557464\, n^8 + 1091213736330860 19\, n^{11} + 26874700372746516\, n^{12}$$
$$+ 6500512066104\, n^{16} + 84729238051860\, n^{15} + 5194295369065098\, n^{13}$$
$$+ 2514511550318768 0 + 2110386357125994 24\, n)\, (n + 3)^3\, \mathrm{s}(n + 3) + (54585830156$$
$$+ 350689467812\, n + 1017700462466\, n^2 + 1760584594380\, n^3 + 2017065459849\, n^4$$
$$+ 1606745735736\, n^5 + 907992479736\, n^6 + 364013859042\, n^7 + 101460307545\, n^8$$
$$+ 18726925518\, n^9 + 2060304120\, n^{10} + 102375360\, n^{11})\, (n + 3)^3\, (n + 4)^7\, \mathrm{s}(n + 4) + 8$$
$$(n + 2)\, (7072908871680\, n^{20} + 315628558398720\, n^{19} + 6650661243415104\, n^{18}$$
$$+ 87979206823913808\, n^{17} + 819439991165553516\, n^{16} + 5711991395289139404\, n^{15}$$
$$+ 30917972174651220597\, n^{14} + 133070276638133809227\, n^{13}$$
$$+ 462516691604036543940\, n^{12} + 1311025295092282143740\, n^{11}$$
$$+ 3047209515781789762641\, n^{10} + 5817899143713103665172\, n^9$$
$$+ 9108545400676905550771\, n^8 + 11630327275776577718556\, n^7$$
$$+ 11993481346952514494264\, n^6 + 9835369404711553127321\, n^5$$
$$+ 6263916978444480644973\, n^4 + 2986089280124489341048\, n^3$$
$$+ 1002446238942897024570\, n^2 + 211318335235609832268\, n$$
$$+ 21039060801453294600)\, \mathrm{s}(n + 1) = 0$$

Four different representations of the Legendre polynomials:

(a) We consider the summand:

```
> legendre1:=binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k;
```

$$legendre1 := \mathrm{binomial}(n, k)\, \mathrm{binomial}(-n - 1, k) \left(\frac{1}{2} - \frac{x}{2}\right)^k$$

The sum

```
> Sum(legendre1,k=0..n);
```

$$\sum_{k=0}^{n} \mathrm{binomial}(n, k)\, \mathrm{binomial}(-n-1, k) \left(\frac{1}{2}-\frac{x}{2}\right)^{k}$$

has the hypergeometric representation

```
> Sumtohyper(legendre1,k);
```

$$\mathrm{Hypergeom}\left([n+1, -n], [1], \frac{1}{2}-\frac{x}{2}\right)$$

and satisfies the recurrence equation

```
> sumrecursion(legendre1,k,P(n));
```

$$(n+1)\, \mathrm{P}(n) - (2\, n+3)\, x\, \mathrm{P}(n+1) + (n+2)\, \mathrm{P}(n+2) = 0$$

(b) We consider the summand:

```
> legendre2:=1/2^n*binomial(n,k)^2*(x-1)^(n-k)*(x+1)^k;
```

$$legendre2 := \frac{\mathrm{binomial}(n, k)^2\, (-1+x)^{(n-k)}\, (x+1)^{k}}{2^{n}}$$

The sum

```
> Sum(legendre2,k=0..n);
```

$$\sum_{k=0}^{n} \frac{\mathrm{binomial}(n, k)^2\, (-1+x)^{(n-k)}\, (x+1)^{k}}{2^{n}}$$

has the hypergeometric representation

```
> Sumtohyper(legendre2,k);
```

$$\frac{(-1+x)^{n}\, \mathrm{Hypergeom}\left([-n, -n], [1], \frac{x+1}{-1+x}\right)}{2^{n}}$$

and satisfies the recurrence equation

```
> sumrecursion(legendre2,k,P(n));
```

$$(n+1)\, \mathrm{P}(n) - (2\, n+3)\, x\, \mathrm{P}(n+1) + (n+2)\, \mathrm{P}(n+2) = 0$$

(c) We consider the summand:

```
> legendre3:=1/2^n*(-1)^k*binomial(n,k)*binomial(2*n-2*k,n)*x^(
  n-2*k);
```

$$legendre3 := \frac{(-1)^{k}\, \mathrm{binomial}(n, k)\, \mathrm{binomial}(2\, n-2\, k, n)\, x^{(n-2\, k)}}{2^{n}}$$

The sum

```
> Sum(legendre3,k=0..floor(n/2));
```

$$\sum_{k=0}^{\mathrm{floor}\left(\frac{n}{2}\right)} \frac{(-1)^{k}\, \mathrm{binomial}(n, k)\, \mathrm{binomial}(2\, n-2\, k, n)\, x^{(n-2\, k)}}{2^{n}}$$

has the hypergeometric representation

```
> Sumtohyper(legendre3,k);
```

$$\frac{\Gamma(2n+1)\,x^n\,\mathrm{Hypergeom}\left(\left[-\frac{n}{2}, -\frac{n}{2}+\frac{1}{2}\right], \left[-n+\frac{1}{2}\right], \frac{1}{x^2}\right)}{2^n\,\Gamma(n+1)^2}$$

and satisfies the recurrence equation

```
> sumrecursion(legendre3,k,P(n));
```

$$(n+1)\,\mathrm{P}(n)-(2n+3)\,x\,\mathrm{P}(n+1)+(n+2)\,\mathrm{P}(n+2)=0$$

(d) We consider the summand:

```
> legendre4:=x^n*hyperterm([-n/2,(1-n)/2],[1],1-1/x^2,k);
```

$$legendre4 := \frac{x^n\,\mathrm{pochhammer}\left(-\frac{n}{2}, k\right)\mathrm{pochhammer}\left(-\frac{n}{2}+\frac{1}{2}, k\right)\left(1-\frac{1}{x^2}\right)^k}{(k!)^2}$$

The sum

```
> Sum(legendre4,k=0..floor(n/2));
```

$$\sum_{k=0}^{\mathrm{floor}\left(\frac{n}{2}\right)} \frac{x^n\,\mathrm{pochhammer}\left(-\frac{n}{2}, k\right)\mathrm{pochhammer}\left(-\frac{n}{2}+\frac{1}{2}, k\right)\left(1-\frac{1}{x^2}\right)^k}{(k!)^2}$$

has the hypergeometric representation

```
> Sumtohyper(legendre4,k);
```

$$x^n\,\mathrm{Hypergeom}\left(\left[-\frac{n}{2}, -\frac{n}{2}+\frac{1}{2}\right], [1], \frac{(-1+x)(x+1)}{x^2}\right)$$

and satisfies the recurrence equation

```
> sumrecursion(legendre4,k,P(n));
```

$$(n+1)\,\mathrm{P}(n)-(2n+3)\,x\,\mathrm{P}(n+1)+(n+2)\,\mathrm{P}(n+2)=0$$

```
>
```

Proof of Clausen's formula by Cauchy product:

```
> summand:=j->hyperterm([a,b],[a+b+1/2],1,j);
```

$$summand := j \to \mathrm{hyperterm}\left([a, b], \left[a+b+\frac{1}{2}\right], 1, j\right)$$

```
> Closedform(summand(j)*summand(k-j),j,k);
```

$$\mathrm{Hyperterm}\left([2b, 2a, a+b], \left[2b+2a, a+b+\frac{1}{2}\right], 1, k\right)$$

Proof of Clausen's formula by differential equations:

The left hand factor satisfies the differential equation

```
> DE:=sumdiffeq(summand(j)*x^j,j,C(x));
```

$DE :=$

$$2(-1+x)\,x\left(\frac{d^2}{dx^2}\mathrm{C}(x)\right)+(2xa-1-2a-2b+2xb+2x)\left(\frac{d}{dx}\mathrm{C}(x)\right)+2\,\mathrm{C}(x)\,b\,a=0$$

Therefore the left hand side satisfies the differential equation

```
> with(gfun):
> LHS:=`diffeq*diffeq`(DE,DE,C(x));
```

$LHS := (8a^2b+8b^2a)\,\mathrm{C}(x)+$

$$(6\,x\,a+4\,x\,b^2+6\,x\,b+16\,b\,a\,x+2\,x+4\,x\,a^2-2\,b-4\,b^2-8\,b\,a-2\,a-4\,a^2)\left(\frac{d}{dx}\mathrm{C}(x)\right)+(6\,x^2\,a+6\,x^2\,b+6\,x^2-6\,x\,a-6\,x\,b-3\,x)\left(\frac{d^2}{dx^2}\mathrm{C}(x)\right)+(-2\,x^2+2\,x^3)\left(\frac{d^3}{dx^3}\mathrm{C}(x)\right)$$

On the other hand the right hand side satisfies the differential equation

```
> RHS:=sumdiffeq(hyperterm([2*a,2*b,a+b],[2*a+2*b,a+b+1/2],x,k)
,k,C(x));
```

$$RHS := 8\,\mathrm{C}(x)\,b\,a\,(a+b)+2\,(2\,x\,b^2+2\,x\,a^2+8\,b\,a\,x+x-2\,a^2-2\,b^2-a-b+3\,x\,b-4\,b\,a+3\,x\,a)\left(\frac{d}{dx}\mathrm{C}(x)\right)+3\,x\,(2\,x\,a-1-2\,a-2\,b+2\,x\,b+2\,x)\left(\frac{d^2}{dx^2}\mathrm{C}(x)\right)+2\,(-1+x)\,x^2\left(\frac{d^3}{dx^3}\mathrm{C}(x)\right)=0$$

These are equal:

```
> expand(LHS-op(1,RHS));
```

$$0$$

```
>
```